包名 结绳.Windows

@全局类
类 杂类操作
	/*
	判断当前应用打包版本是否为调试版
	*/
	@静态
	方法 为调试版() : 逻辑型
		code return #[debug];
	结束 方法
	
	/*
	获取当前应用的构建时间文本
	*/
	@静态
	方法 取构建时间() : 文本
		code return #[date];
	结束 方法
	
	/*
	获取当前应用的构建时间戳
	*/
	@静态
	方法 取构建时间戳() : 长整数
		code return #[time];
	结束 方法
	
	/*
	获取当前代码所处源代码行号
	*/
	@静态
	@嵌入式代码
	方法 取当前行号() : 整数
		code #[line]
	结束 方法
	
	/*
	获取当前代码所处源文件路径
	*/
	@静态
	@嵌入式代码
	方法 取当前源文件路径() : 文本
		code #[source]
	结束 方法
	
	/*
	取反
	*/
	@静态
	方法 取反(值 : 逻辑型) : 逻辑型
		如果 值 == 真 则
			返回 (假)
		否则
			返回 (真)
		结束 如果
	结束 方法
	
	/*
	为指定变量进行赋值
	*/
	@静态
	@嵌入式代码
	方法 赋值(变量名 : 变体型, 值 : 变体型)
		code #变量名 = #值
	结束 方法
	
	/*
	将指定数字自增
	参数一: 欲操作数字
	参数二: 欲增加的值，默认为1
	*/
	@静态
	@嵌入式代码
	方法 自增(数字 : 变体型, 自增值 : 变体型 = 1)
		code #数字 += #自增值
	结束 方法
	
	/*
	将指定数字自减
	参数一: 欲操作数字
	参数二: 欲减去的值，默认为1
	*/
	@静态
	@嵌入式代码
	方法 自减(数字 : 变体型, 自减值 : 变体型 = 1)
		code #数字 -= #自减值
	结束 方法
结束 类